NCL 绘图示例（三）：小波图

施宁
（南京信息工程大学 大气科学学院）




```
begin

  f       = addfile("./ENSO-index.nc", "r")
  ensoi   = f->ensoi

  time = ensoi&year

  N   = dimsizes(ensoi)

  ;; 小波计算
  mother   = 0          ; 母小波类型，通常为 0，即'Morlet'小波。其余两中被分别为 1，'Paul'小波和 2，'DOG' (derivative of Gaussian)小波
  dt       = 1          ; 数组中数值之间的时间间隔，通常为 1。本例中表示间隔 1 年。
  param    = -1         ; 母小波参数。 如果 param < 0，则使用默认数值，即采用'Morlet'小波时为 6；Paul'小波为 4;'DOG'小波为 2
  s0       = dt         ; 'Morlet'小波 s0 = dt ; 'Paul'小波 s0 = dt/4
  dj       = 0.25       ; 常用设定
  jtot     = 1+floattointeger(((log10(N*dt/s0))/dj)/log10(2.)) ; 常用设定
  npad     = N          ; 常用设定
```

```
  nadof     = 0          ; 常用设定
  noise     = 1          ; 常用设定,h 红噪声检验
  siglvl    = .05        ; 置信度水平
  isigtest=  0           ; 采用 chi-square 检验；若为 1 则是对全部波谱进行时间平均检验

  w = wavelet(ensoi,mother,dt,param,s0,dj,jtot,npad,noise,isigtest,siglvl,nadof)

  ;************************************
  power                 = onedtond(w@power,(/jtot,N/))   ; 功率谱
  power!0               = "period"                       ; Y axis
  power&period          = w@period

  power!1               = "time"                         ; X axis
  power&time            = time

  power@long_name    = "Power Spectrum"
  power@units           = "1/unit-freq"

  ;计算显著性 ( >= 1 则显著)
  SIG                   = power                          ; 复制元数据
  SIG                   = power/conform (power,w@signif,0)
  SIG@long_name       = "Significance"
  SIG@units             = " "

;************************************************
  wks = gsn_open_wks("eps","plot-enso-wavelet")
  gsn_define_colormap(wks,"BlAqGrYeOrReVi200")

  YLValues = (/1,2,4,8,16/)
  YLLabels = (/"1","2","4","8","16"/)

  res                           = True
  res@gsnDraw                   = False
  res@gsnFrame                  = False
  res@gsnRightString          = " "
  res@gsnLeftString           = " "

  res@trYReverse = True                         ; 倒置 y-axis
  res@tmYLMode      = "Explicit"
  res@tmYLValues = YLValues
  res@tmYLLabels = YLLabels
  res@tmLabelAutoStride      = True
```

```
  res@trYMaxF              = max(YLValues)
  ;res@trYMinF              = min(YLValues)

  res@cnLinesOn           = False
  res@cnLineLabelsOn = False
  res@cnInfoLabelOn    = False

  res2 = res

  res@tiXAxisString            = "Year"
  res@tiXAxisOffsetYF         = 0.135
  res@tiYAxisString            = "Years"
  res@cnFillOn                   = True
  res@cnFillMode                = "RasterFill"
  res@cnRasterSmoothingOn = True

  ;;;;;;;;;;;;
  res2@cnLevelSelectionMode = "ManualLevels"
  res2@cnMinLevelValF            = 0.00
  res2@cnMaxLevelValF            = 2.00
  res2@cnLevelSpacingF          = 1.00
  res2@cnFillScaleF               = 0.5        ; 增加形状填充的密度（通过下面调用
ShadeGtContour 实现形状填充）

  plot = gsn_csm_contour(wks,power,res)
  iplot = gsn_csm_contour(wks,SIG,res2)

  opt = True
  opt@gsnShadeFillType = "pattern"     ; 默认设置
  opt@gsnShadeHigh       = 17 ;见附录图 A.3
  iplot  =  gsn_contour_shade(iplot,-999.,1.,opt) ; 从大于等于 1.的第一个等值线开
始用形状为 17 填充
  overlay(plot,iplot)                  ; 在原图上添加显著性

  plot = ShadeCOI(wks,plot,w,time,False) ;

  ;;添加各频率的功率
  gws = w@gws
  resl = True
  resl@gsnFrame          = False
  resl@gsnDraw           = False
  resl@trYAxisType      = "LogAxis"
  resl@trYReverse       = True                          ; reverse y-axis
  resl@tmYLMode          = "Explicit"
```

```
   resl@tmYLValues        = YLValues
   resl@tmYLLabels        = YLLabels
   resl@trYMaxF            = max(YLValues)
   resl@trYMinF            = min(YLValues)
   resl@tiXAxisString = "Global Wavelet Power"

   plotg = gsn_csm_xy(wks,gws,power&period,resl)

   ;;  将 plotg 添加至 plot 的右侧
   plotc = gsn_attach_plots(plot,plotg,res,resl)

  draw(plot)
  frame(wks)
end
```